
条码打印机

ZMIN PCLE 指令手册

Version: 1.0.0.3

2016
深圳致明兴科技有限公司



目 录

PCLE 指令	1
B1. 指令格式	2
1. 基本指令格式	2
2. 字符串	2
B2. 字体	3
1. 内部字体	3
2. 软字体	3
3. 符号集	3
B3. 指令集汇总	5
条码打印机的坐标系统	6
提示	7
T 命令: 打印一个文本字符串,序列号,变量	8
B 命令: 打印一个条码	10
b 命令: 打印二维条码	13
BI 命令: 打印 Bin 图形名称清单	19
BK 命令: 删除一个或所有的 Bin 图形	20
C 命令: 设置一个序列号变量	21
H 命令: 设置打印黑度	23
EI 命令: 打印软字体名称清单	24
EK 命令: 删除一个或全部的软字体	25
ES 命令: 下载一个软字体到打印机的 RAM 或 FLASH	26
FM 命令: 命令打印机走一行标签	28
FE 命令: 结束存储表格(Form)	29
FI 命令: 打印存储在打印机里的表格名称清单	30
FK 命令: 删除存储在打印机里的一个或所有的表格	31
FR 命令: 运行一个表格	32
FS 命令: 存储一个表格到打印机	33
GC 命令: 打印一个 Bin 格式图形	34
GD 命令: 存储一个 Bin 格式的图形到打印机	35
GG 命令: 打印一个图形	36
GI 命令: 打印图形名称清单	37
GK 命令: 删除一个或所有的图形	38
GM 命令: 存储一个图形到打印机	39
GW 命令: 打印二进制格式图形	40
I 命令: 选择字符集	41
JB/JF 命令: 取消/设置打印回转功能	42
LE 命令: 画直线(两直线相交处作”异或”处理)	43
LO 命令:画直线(两直线相交处作”或”处理)	44

LS 命令:画斜线	45
LW 命令: 画白色直线.....	46
MD 命令: 打印机测纸	47
N 命令: 清除打印机缓冲内存的内容	48
O 命令: 设置打印机的工作状态	49
W 命令: 打印标签.....	50
WA 命令: 自动打印标签(只能在 FORM 里使用).....	51
Q 命令: 设置标签的高度和定位间隙\黑线\穿孔的高度	52
q 命令: 设置标签的宽度.....	53
R 命令: 设置/改变坐标原点	54
S 命令: 设置打印机打印速度.....	55
U 命令: 打印机器的设置状态	56
UN/US 命令: 取消/设置错误反馈.....	58
V 命令: 定义变量.....	59
X 命令: 画矩形.....	59
Z 命令: 设置标签打印方向	61
ZN/ZS 命令: 取消/选择 FLASH 存储器	62
?命令: 下载变量或序列号变量	63
^@命令: 将打印机复位	64
^ee 命令: 要求打印机立刻反馈错误报告.....	65
TD 命令: 设置打印日期格式.....	66
TS 命令: 设置打印机内部 RTC 日期和时间.....	67
TT 命令: 设置打印时间格式.....	68

PCLE 指令

版本编号：1.X.X.X

版权所有：©2012~深圳致明兴科技有限公司。保留所有权利。

PCLE（Printer command language E）为深圳致明兴科技有限公司 ZMIN 条码打印机的用户提供一组控制指令。

此指令集仅支持本公司产品。

B1. 指令格式

- 指令名称一般由 1 至 2 个字母组成。(区分大小写)
- 有些指令需一个或多个参数，有些指令需可选的参数以实现特别的功能。
- 所有指令行必须以一个回车换行 (<CR> - 0DH) 控制码作为结束标记。

1. 基本指令格式

* 格式 1: 无参数的指令

语法	描述
X<CR>	单字符指令
XX<CR>	双字符指令

* 格式 2: 固定参数数目的指令

语法	描述
Xp1, p2, p3, ..., pn <CR>	单引导字符指令
XXp1, p2, p3, ..., pn <CR>	双引导字符指令

* 格式 3: 可选参数的指令

X[p1, p2, p3, ..., pn] <CR>

2. 字符串

* 字符串以双引号 (") 作为起始和结束标记。

* 反斜杠 (\) 有以下作用:

字符	输入
"	\ "
\	\\
0x00 - 0x7F	\x00 - \x7F

注：所有打印指令和名称均请注意区分大小写。

B2. 字体

根据存储周期的不同，打印指令集定义了 2 种字体：

- 内部字体
- 软字体 (SoftFont)

1. 内部字体

有 5 种内部字体被存储在打印机中，与软字体 (SoftFont) 不同，它们不能被用户删除。
每种字体各自有一个唯一的 ID 编号。

ID 号	203DPI	300DPI	注释
1	8 x 12 dots, 字间距 2dots	12 x 20 dots, 字间距 2dots	
2	10 x 16 dots, 字间距 2dots	16 x 28 dots, 字间距 2dots	
3	12 x 20 dots, 字间距 2dots	20 x 36 dots, 字间距 3dots	
4	14 x 24 dots, 字间距 2dots	24 x 44 dots, 字间距 3dots	
5	32 x 48 dots, 字间距 3dots	48 x 80 dots, 字间距 3dots	不包含小写字母字符
6	24 x 24 dots, 字间距 2dots	24 x 24 dots, 字间距 2dots	仅支持中文
a	24 x 24 dots, 字间距 2dots	24 x 24 dots, 字间距 2dots	支持中文及大小写字母、数字、符号等

2. 软字体

软字体由用户使用指令或应用软件从主机下载至打印机。

如果内部字体不能满足需求，软字体是很好的解决办法。

可根据需要下载指定数量的字符。

每种软字体各自都有一个唯一的 ID 编号，根据此 ID 编号，软字体可被下载、选择或删除。

软字体 ID 编号从 A 至 Z。

3. 符号集

代码表可被重新定义为另一种符号集或代码页。

	8 比特字符	7 比特字符
符号集	Code page 437 Code page 850	USASCII, British, Danish, French,

	Code page 852 Code page 860 Code page 863 Code page 865	German, Italian, Spanish, Swedish, Swiss
--	--	--

B3. 指令集汇总

本指令集含如下类别：


- 设置指令
- 标签格式指令
- 通过串口的交互指令
- 对象下载指令

索引

指令	描述	指令	描述
T	打印文本	LE	“异或”画线
B	打印一维条码	LO	“或”画线
b	打印二维条码	LW	画白线
C	计数器 (counter)	N	清除图象缓冲区 (frame)
H	设置打印黑度	O	选择选项**
EI	打印软字体名称	W	打印标签
EK	删除软字体	WA	自动打印
ES	下载软字体	Q	设定标签及间隙长度**
FE	结束表格存储 (form)	q	设定标签宽度**
FI	打印表格名称	R	设置原点位置 (移动坐标 0 点) **
FK	删除表格	S	设置打印速度**
FR	执行表格	TD	设置日期格式
FS	表格存储	TS	设置 RTC 日期及时间
GG	打印图形	TT	设置时间格式
GI	打印图形列表 (list)	U	打印机器设置状态
GK	删除图形	UN	设置错误报告无效
GM	存储图形	US	设置错误报告有效
GW	打印二进制格式图形	V	定义变量
I	选择字符集**	X	画矩形
JB	设置回转功能无效**	Z	设置打印方向
JF	设置回转功能有效**	ZS	设置-存储至闪存
FM	命令打印机走一行标签	ZN	设置-存储至动态存储器
MD	校准纸张感应器	?	下载变量
		^ee	要求打印机立刻反馈错误报告
		^@	将打印机复位

条码打印机的坐标系统

如下图所示：



提示

- * <CR>为 USASCII 码十进制的” 13” , 或十六进制的” 0DH” , 即” 回车” 。
- * 点(dots)等于一英寸除以打印机的最大分辨率:
 - 对于 203DPI 的打印机来说, $1\text{dot} = 25.4\text{mm}/203 = 0.125\text{mm}(1\text{dot}=1000/203=5\text{mil})$;
 - 对于 300DPI 的打印机来说, $1\text{dot} = 25.4\text{mm}/300 = 0.085\text{mm}(1\text{dot}=1000/300=3\text{mil})$ 。

T 命令: 打印一个文本字符串,序列号,变量

语法:

```
Tp1, p2, p3, p4, p5, p6, p7, " DATA" <CR>
Tp1, p2, p3, p4, p5, p6, p7, Cn<CR>
Tp1, p2, p3, p4, p5, p6, p7, Vn<CR>
Tp1, p2, p3, p4, p5, p6, p7, " DATA" Cn<CR>
Tp1, p2, p3, p4, p5, p6, p7, " DATA" Vn<CR>
Tp1, p2, p3, p4, p5, p6, p7, Combo <CR>
```

参数说明:

- p1: 设置 X 坐标, 以点(dots)为单位。
- p2: 设置 Y 坐标, 以点(dots)为单位。
- p3: 选择打印方向. 0—不旋转; 1—旋转 90° ; 2—旋转 180° ; 3—旋转 270° 。
- p4: 选择内置字体或软字体。
 - 1—5: 为打印机内部英文字体;
 - 6 为打印机内置 24*24 宋体汉字 (仅显示中文);
 - a 为打印机内置 24*24 宋体汉字 (可显示中文、英文、数字符号等);
 - A—Z: 为用户下载的软字体。

取值	描述
1	西文字体1
2	西文字体2
3	西文字体3
4	西文字体4
5	西文字体5
6	24点阵中文字体, 仅中文
a	24点阵中文字体, 中文、英文、数字符号等
A~Z	软字体

- p5: 设置文字的水平放大系数, 可选择:1—24。
- p6: 设置文字的垂直放大系数, 可选择:1—24。
- p7: 选' N' 则打印正常文本(如白底黑字文本),
选' R' 则打印文本反色文本(如黑底白字文本)。
- “DATA”: 一个常量字符串。
- Cn: 一个序列号数值. 请参考 C 命令。
- Vn: 一个变量字符串. 请参考 V 命令。
- Combo: 组合字符串。

用户可以用 " DATA"、Cn、Vn 自由排列组合成一个组合字符串, 如: "data1" CnVn
"data2"。

注: 字符串的总长度不能超过 100 个字符。

例:

```
N
T50, 56, 0, 1, 1, 1, N, "FONT1:ABCabc012"
T50, 72, 0, 2, 1, 1, N, "FONT2:ABCabc012"
T50, 96, 0, 3, 1, 1, N, "FONT3:ABCabc012"
T50, 128, 0, 4, 1, 1, N, "FONT4:ABCabc012"
T50, 168, 0, 5, 1, 1, N, "FONT5:ABC012"
W1
```

打印输出:

```
FONT1:ABCabc012
FONT2:ABCabc012
FONT3:ABCabc012
FONT4:ABCabc012
FONT5:ABC012
```

B 命令: 打印一个条码

语法:

Bp1, p2, p3, p4, p5, p6, p7, p8, " DATA" <CR>
 Bp1, p2, p3, p4, p5, p6, p7, p8, Cn<CR>
 Bp1, p2, p3, p4, p5, p6, p7, p8, Vn <CR>
 Bp1, p2, p3, p4, p5, p6, p7, p8, " DATA" Cn <CR>
 Bp1, p2, p3, p4, p5, p6, p7, p8, " DATA" Vn<CR>
 Bp1, p2, p3, p4, p5, p6, p7, Combo <CR>

参数说明:

p1: 设置 X 坐标, 以点(dots)为单位。
 p2: 设置 Y 坐标, 以点(dots)为单位。
 p3: 选择打印方向。0—不旋转; 1—旋转 90°; 2—旋转 180°; 3—旋转 270°。
 p4: 选择要打印的条码码制。

P4 值	条码码制
0	Code 128 UCC (shipping container code)
1	Code 128 AUTO
1A	Code 128 subset A
1B	Code 128 subset B
1C	Code 128 subset C
1E	UCC/EAN
2	Interleaved 2 of 5
2C	Interleaved 2 of 5 with check sum digit
2D	Interleaved 2 of 5 with human readable check digit
2G	German Postcode
2M	Matrix 2 of 5
2U	UPC Interleaved 2 of 5
3	Code 3 of 9
3C	Code 3 of 9 with check sum digit
3E	Extended Code 3 of 9
3F	Extended Code 3 of 9 with check sum digit
9	Code93
E30	EAN-13
E32	EAN-13 2 digit add-on
E35	EAN-13 5 digit add-on
E80	EAN-8
E82	EAN-8 2 digit add-on
E-85	EAN-8 5 digit add-on
K	Codabar
P	Postnet
UA0	UPC-A
UA2	UPC-A 2 digit add-on
UA5	UPC-A 5 digit add-on

UE0	UPC-E
UE2	UPC-E 2 digit add-on
UE5	UPC-E 5 digit add-on

p5: 设置窄条码的宽度, 以点(dots)为单位。
 p6: 设置宽条码的宽度, 以点(dots)为单位。
 p7: 设置条码高度, 以点(dots)为单位。
 P8: 选' N' 则打印正常文本(如白底黑字文本),
 选' R' 则打印文本反色文本(如黑底白字文本)。
 “DATA” : 一个常量字符串。
 Cn: 一个序列号数值. 请参考 C 命令。
 Vn: 一个变量字符串. 请参考 V 命令。
 Combo: 组合字符串. 用户可以用” DATA”、Cn、Vn 自由排列组合成一个组合字符串,
 如: “data1” CnVn “data2”。

注: 字符串的总长度不能超过 100 个字符。

例:
 N
 B20, 20, 0, E80, 3, 3, 41, B, "0123459"
 B20, 120, 0, K, 3, 5, 61, B, "A0B1C2D3"
 B190, 300, 2, 1, 2, 2, 51, B, "0123456789"
 B20, 330, 0, UA0, 2, 2, 41, B, "13579024680"
 W1

打印输出:



b 命令: 打印二维条码

语法:

bp1, p2, p3, [specific parameters and data]

参数说明:

p1: 设置 X 坐标, 以点 (dots) 为单位.

p2: 设置 Y 坐标, 以点 (dots) 为单位.

p3: 选择要打印的码制.

p3 值	条码码制
M	Maxi Code
P	PDF-417
DX	DataMatrix ECC200
D0	DataMatrix ECC000 (*)
D1	DataMatrix ECC050 (*)
D2	DataMatrix ECC080 (*)
D3	DataMatrix ECC100 (*)
D4	DataMatrix ECC140 (*)
QR	QR
HX	汉信码

* 当您想使用 DataMatrix ECC000-ECC140 条码码制时, 请联系我们!

Maxi Code

1、UPS 格式数据:

[Mode,1],”Data”

Mode: 表示模式 2 - 4.

1: 表示将打印 UPS 格式数据.

2、非 UPS 格式数据:

[Mode,0,C1,Co,Po],”Data”

Mode: 表示模式 M2, M3, m4, m6.

0: 表示将打印非 UPS 格式数据.

C1: 3 位数字.

Co: 3 位数字.

Po: 如果当前模式为 3, PostalCode 则为不超过 6 个字符的字符数据; (A - Z, or 0 - 9)

如果当前模式为 2, PostalCode 则为 9 个数字字符数据.

注: RS 控制符用于分开数据流格式由'_1E'表示,GS 控制符用于隔开数据字段由'_1D'表示,EOT 控制符用于表示数据流结束由'_04'表示. 资料容量最大为 93 个字元, 若纯为数字字元, 则可存放 138 个 .

UPS 标准格式如下:

Message Header: []>RS

Transportation Data

Format Header : 01GSyy <其中 yy 表示年份>

Tracking Number: *<tracking number>

SCAC :GS<SCAC>
 UPS Shipper Number :GS<shipper number>
 Julian Day of Pickup: GS<day of pickup>
 Shipment ID Number :GS<shipment ID number>
 Package n/x:GS<n/x>
 Package Weight :GS<weight>
 Address Validation :GS<validation>
 Ship to Street Address :GS<street address>
 Ship to City: GS<city>
 Ship to State: GS<state>
 RS: RS
 End of Message: EOT
 (* Mandatory Data for UPS 必须设置数据)

例:


N

b10, 10, M, m4, 0, 1Z000A7&dajc_iaj-3=+~#\$5&#fe[{: 's;fk., LAKFalkdggjoiu12815/w35::'./dfoi38&*^}

!+

W1

打印输出:



PDF-417

[w,v,s,c,p,x,y,r,l,t,o],”Data”

- w: 最大打印宽度, 以点(dots)为单位.
- v: 最大打印高度, 以点(dots)为单位.
- s: 纠错等级, 范围: 0 - 8.
- c: 数据压缩等级, 范围: 0 or 1. 缺省为 0.
- x: 模块宽度, 范围: 2—9, 以点(dots)为单位.
- y: 模块高度, 范围: 4—99, 以点(dots)为单位.
- r: 最大行数.
- l: 最大列数.
- t: 截取标志, 0=不截取, 1=截取.
- o: 设置旋转方向.

0—0° , 1—90° , 2—180° , 3—270° .

例:

N

b50, 30, P, 00, 00, s0, c0, x3, y6, r0, 10, t0, o0, "ABCabc12345"

W1

打印输出:



Datamatrix Ecc200

[w,v,o,m],"Data"

w: 最大打印宽度, 以点(dots)为单位.

v: 最大打印高度, 以点(dots)为单位.

o: 设置旋转方向.

范围值: (0 - 3)

(0—0° , 1—90° , 2—180° , 3—270°)

m : 设置放大倍数, 以点(dots)为单位

范围值: (1 - 9)

(1—放大 1 倍, 2—放大 2 倍 3—放大 3 倍……)。

例:

N

b20, 220, DX, 0, 0, o0, m5, "Data Matrix"

W1

打印输出:



QR

[w,v,o,r,m,g,s],"data"

w: 最大打印宽度, 以点(dots)为单位.

v: 最大打印高度, 以点(dots)为单位.

o: 设置旋转方向.

范围值(0 - 3)

(0--0° , 1--90° , 2--180° , 3--270°)

r: 设置放大倍数, 以点(dots)为单位.

范围值: (1 - 9)

(1—放大 1 倍, 2—放大 2 倍 3—放大 3 倍……)

m: QR 码编码模式选择.

范围值(0 - 4)

0 是选择数字模式

1 是选择数字字母模式

2 是选择字节模式 0~256

3 是选择中国汉字模式

4 是选择混合模式

g: QR 码纠错等级选择.

范围值(0 - 3)

0 是'L' 等级

1 是'M' 等级

2 是'Q1' 等级

3 是'H1' 等级

s: QR 码掩模图形选择

范围值(0 - 8)

0 - 是掩模图形 000

1 - 是掩模图形 001

2 - 是掩模图形 010

3 - 是掩模图形 011

4 - 是掩模图形 100

5 - 是掩模图形 101

6 - 是掩模图形 110

7 - 是掩模图形 111

8 - 是自动选择掩模图形

例:

N

b200, 200, QR, 0, 0, o0, r5, m2, g0, s0, "ABCabc12345"

W1

打印输出:



汉信码

[w, v, m, o, r, g, s], "data"

w: 最大打印宽度, 以点(dots)为单位.

v: 最大打印高度, 以点(dots)为单位.

m: 汉信码编码模式选择

范围值(0 到 6)

0 是选择数字模式

1 是选择 TEXT 模式

2 是选择二进制模式

3 是选择常用汉字 1 区模式编码

4 是选择常用汉字 2 区模式编码

5 是 GB 18030 双字节区模式

6 是 GB 18030 四字节模式编码

o: 设置旋转方向.

范围值(0 到 3)

(0--0° , 1--90° , 2--180° , 3--270°)

r: 设置放大倍数, 以点(dots)为单位

范围值: (0 - 30)

(0—放大 1 倍, 1—放大 2 倍 2—放大 3 倍……依此类推)。

g: 汉信码纠错等级选择

范围值(0 到 3)

0 是' L1' 等级

1 是' L2' 等级

2 是' L3' 等级

3 是' L4' 等级

s: 汉信码掩模图形选择

范围值(0 到 3)

0 是掩模图形 00

1 是掩模图形 01

2 是掩模图形 10

3 是掩模图形 11

例:

N

b100, 100, HX, 0, 0, m1, o0, r5, g3, s2, "ZMIN"

W1

打印输出:



BI 命令: 打印 Bin 图形名称清单

语法: BI<CR>

说明: 此命令将打印已存储在 RAM 或 FLASH 存储器里的图形的名称, (目前包含 BIN 格式和 PCX 格式的图形)。

例:

BI

打印输出:

如果打印机没有存储图形, 则打印:

**Graphics Information:
NO Graphics Stored**

如果打印机里存储了名称为 TF2, TF3 等的图形, 则打印:

**Graphics Information:
TF2
TF3
TF4
PB5
PB6
DB7
TF9**

BK 命令：删除一个或所有的 Bin 图形

语法：

BK " GNAME" <CR>

BK "*" <CR>

参数说明：

GNAME：图形名称；最大长度为 16 个字符。

*：如果选择这个参数，将删除所有存储在 RAM 或 FLASH 存储器里的图形，（目前包含 BIN 格式和 PCX 格式的图形）。

C 命令: 设置一个序列号变量

语法: Cp1, p2, p3, p4, " MSG" <CR>

参数说明:

p1: 序列号 ID. 取值范围:0—9.

p2: 序列号最大数字个数. 取值范围:1—40.

p3: 对齐标志. L—左对齐;R—右对齐;C—居中;N—不对齐.

p4: 序列号的变化规律. 由" + " 或" - " 加上一个数字, 再加上一个变化标志(D - 十进制, B - 二进制, O - 八进制, H - 十六进制, X-自定义模式, 可以允许用户输入最多64个字符.) 组成:

" +1 " =每次增加 1, 默认按照十进制计算. 如 1234, 1235, 1236,

" +3D " =每次增加 3, 按照十进制计算. 同上.

" -1B " =每次减少 1, 按照二进制计算. 如 1111, 1110, 1101,

" -4O " =每次减少 4, 按照八进制计算. 如 1234, 1230, 1224,

" -6H " =每次减少 6, 按照十六进制计算. 如常 1234, 122E, 1228,

" +3X " =每次内容根据用户自定义的变化规律增加 3.

如变化规律表内容为:TE2DOKLU046MNY37, 起始值是" T062",

则 T062, T06K, T06O,

"MSG": 提示信息字符串. 可显示在打印机显示屏上或可编程键盘(KDU)的显示屏上.

例:

```
N
FK"TEST"
FS"TEST"
C0, 6, N, +1, "Enter Code:"
T100, 100, 0, 4, 1, 1, N, "Label: "
T300, 100, 0, 4, 1, 1, N, C0
FE
```

以上命令将一个 FORM:" TEST" 存储到打印机.

如果将以下命令发送到打印机, 将会打印出内容:

```
FR"TEST"
?
1000
W2
```

打印输出:

Label: 1000

Label: 1001

H 命令: 设置打印黑度

语法: Hp1<CR>

参数说明: p1: 取值范围: 0—20, 缺省为 10. 此值并不是真正意义的温度数值, 而是相对数值, 0 表示打印头工作在最小发热状态, 20 表示打印工作在最大发热状态.

例:

```
N
H10
T100, 100, 0, 3, 1, 1, N, "DARKNESS=10"
W1
```

EI 命令: 打印软字体名称清单

语法: EI<CR>

说明: 此命令要求打印机打印存储在 RAM 或 FLASH 存储器里的软字体的名称清单.

例:

EI

打印输出:

如果打印机里没有存储软字体, 将打印:

**Soft Font Information:
NO Soft Font Stored**

如果有软字体存储在打印机上, ID 分别为 C, D, E, F, G, 则打印:

```
Soft Font Information:  
C  
D  
E  
F  
G
```

EK 命令: 删除一个或全部的软字体

语法:

EK " ID" <CR>

EK " *" <CR>

说明: 此命令要求打印机删除存储在 RAM 或 FLASH 存储器里的一个或所有的软字体.

参数说明:

ID: 软字体 ID, 取值范围:A—Z.

*: 如果选择这个参数, 打印机将删除存储在 RAM 或 FLASH 存储器里所有的软字体.

ES 命令：下载一个软字体到打印机的 RAM 或 FLASH

语法: ES” ID” p1,p2,...

参数说明:

- ID: 一个大写字母 A—Z.
- p1: 将要下载软字体的字符个数.
- p2: 字符高度.

基本的软字体格式为:

字体格式 0
字符 0
...
字体格式 N-1
字符 N-1

 :

“第 N 个字符” 的字体格式:


字节 0, 1	该字符的十六进制代码. 对于 ASCII 字符, 高字节为 0x00, 低字节为 ASCII 代码; 对于汉字, 为国标编码.
字节 2	字符总宽度, 等于字符实际宽度+字符间隔. 以像素(pixels)为单位, 范围 0x00-0xFF.
字节 3	字符实际宽度. 以字节(bytes)为单位, 范围 0x00-0xFF.

:

“第 N 个字符” 的数据格式:

字节 0...	字符图像的数据, 长度为字节 3*p2
---------	---------------------

例:



Parameter	Data	Hexadecimal
字节2	16	10
字节3	14	0E

p2	16	10
----	----	----

FM 命令: 命令打印机走一行标签

语法: FM<CR>

说明: 此命令与按一下 feed 键功能相同

例:

FM

N

T20, 110, 0, 4, 2, 2, N, "LIFE"

W1

FE 命令: 结束存储表格 (Form)

语法: FE<CR>

说明: 此命令与 FS 命令一起使用.

例:

FS” Standalone”

...

FE

FI 命令: 打印存储在打印机里的表格名称清单

语法: FI<CR>

例:

FI

打印输出:

如果打印机里没有存储表格, 将打印:

Form Information:
NO Form Stored

如果打印机里已经存储有名称为 Standalone 的表格, 则打印:

Form Information:
Standalone

FK 命令: 删除存储在打印机里的一个或所有的表格

语法: FK " FORMNAME " <CR>

FK " * " <CR>

参数说明:

FORMNAME: 将要删除的软字体的名称, 最大长度为 16 个字符.

*: 如果选择这个参数, 打印机将删除存储在 RAM 或 FLASH 存储器里所有的表格.

FR 命令: 运行一个表格

语法: FR" FORMNAME" <CR>

参数说明:

FORMNAME: 将要运行的表格的名称;最大长度为 16 个字符.

例:

FK" Standalone"	; 删除名称为 "Standalone" 的表格
FS" Standalone"	; 开始下载新的表格
T50, 30, 0, 4, 1, 1, N, "THIS IS Standalone."	
FE	; 下载新的表格结束
FR" Standalone"	; 执行表格里的命令
W1	

FS 命令：存储一个表格到打印机

语法：FS” FORMNAME” <CR>

说明：此命令与 FE 命令一起使用。

如果使用 ZS 命令，表格的内容则存储到 FLASH 存储器；如果使用 ZN 命令，表格的内容则存储到 RAM 存储器。

参数说明：

FORMNAME：自定义的表格名称；最大长度为 16 个字符。

GC 命令: 打印一个 Bin 格式图形

语法:

GCp1, p2, " GNAME" <CR>

参数说明:

p1: 设置 X 坐标;以点(dots)为单位.

p2: 设置 Y 坐标;以点(dots)为单位.

GNAME: 图形名称;最大长度为 16 个字符;

GD 命令: 存储一个 Bin 格式的图形到打印机

语法:

GD"GNAM" P1,P2, [...raster data...]

参数说明:

GNAM: 自定义图形的名称;最大长度为 16 个字符.

P1: 每行数据的字节数(1Byte=8bits).

P2: 图形的高度;以点(dots)为单位.

P3([...raster data...]): 二进制图形数据, 每一个比特(bit)


表示一个点;比特值为 1 时此点将打印, 为 0 时此点不打印. (与 GW 定义相反)

数据量大小=p1 * p2(Bytes).

二进制数据传输顺序是从左到右, 从上到下, 以下图为例:

数据传输顺序为: Line1 的 Byte1(0x00), Line1 的 Byte2(0x00), Line2 的 Byte1(0x1f), Line2 的 Byte2(0xe0), Line3 的 Byte1(0x00), Line3 的 Byte2(0x00), ...

其中虚线部分是非图形区域, 对应它们的 bit 值为 0。



GG 命令: 打印一个图形

语法: GGp1, p2, " GNAME" <CR>

参数说明:

p1: 设置 X 坐标;以点(dots)为单位.

p2: 设置 Y 坐标;以点(dots)为单位.

GNAME: 图形名称;最大长度为 16 个字符;或者是变量名称(Variable Data:V00-V99).

例:

N

GG100, 50, " PCXGRAPH"

W1

GI 命令: 打印图形名称清单

语法: GI<CR>

说明: 此命令将打印已存储在 RAM 或 FLASH 存储器里的图形的名称.

例:

GI

打印输出:

如果打印机没有存储图形, 则打印:

**Graphics Information:
NO Graphics Stored**

如果打印机里存储了名称为 TF2, TF3 等的图形, 则打印:

**Graphics Information:
TF2
TF3
TF4
PB5
PB6
DB7
TF9**

GK 命令: 删除一个或所有的图形

语法:

GK " GNAME " <CR>

GK " * " <CR>

参数说明:

GNAME: 图形名称;最大长度为 16 个字符.

*

: 如果选择这个参数, 将删除所有存储在 RAM 或 FLASH 存储器里的图形.

GM 命令: 存储一个图形到打印机

语法:

```
GM" GNAME" p1<CR>
PCX file
```

参数说明:

GNAME: 自定义图形的名称;最大长度为 16 个字符.

p1: PCX 文件的大小;一个十进制数值;以字节(Bytes)为单位.

PCX file: 一个 PCX 格式图形.

例:

```
GK" PB6"
GM" PB6" 2086
...[PCX file for PCXA graphics]...
N
T30,30,0,4,1,1,N,"PB6:"
GG30,100,"PB6"
W1
GK"*"
```

打印输出:

PB6:



GW 命令: 打印二进制格式图形


语法: GWp1, p2, p3, p4[...raster data...]<CR>

说明: 必须逐行发送不压缩的图形数据; 每一个比特(bit)表示一个点; 比特值为 0 时此点将打印, 为 1 时此点不打印.

参数说明:

- p1: 设置 X 坐标, 以点(dots)为单位.
- p2: 设置 Y 坐标, 以点(dots)为单位.
- p3: 每行数据的字节数 (1Byte=8bits).
- p4: 图形的高度; 以点(dots)为单位.
- P5([...raster data...]): 二进制图形数据, 数据量大小=p3 X p4(Bytes).

二进制数据传输顺序是从左到右, 从上到下, 以下图为例:
数据传输顺序为: Line1 的 Byte1(0xff), Line1 的 Byte2(0xff), Line2 的 Byte1(0xe0), Line2 的 Byte2(0x1f), Line3 的 Byte1(0xff), Line3 的 Byte2(0xff), ...
其中虚线部分是非图形区域, 对应它们的 bit 值为 1。



I 命令: 选择字符集

语法: Ip1, p2, p3<CR>

说明: 缺省为 Code page 437 (English).

参数说明:

p1: 数据比特值; 8 表示 8 位码, 7 表示 7 位码.

p2: 字符集.

p3: 可编程键盘 (KDU) 的国家编码.

8 位码 (p1=8)	字符集 (Code page)	7 位码 (p1=7)	字符集
0	English (437)	0	USASCII
1	Latin 1 (850)	1	British
2	Slavic (852)	2	German
3	Portugal (860)	3	French
4	Canadian/French (863)	4	Danish
5	Nordic (865)	5	Italian
		6	Spanish
		7	Swedish
		8	Swiss

JB/JF 命令: 取消/设置打印回转功能

语法: JB<CR>

JF[p1]<CR>

说明: 出厂设置为不回转状态.

p1: 回转距离, 以点(dots)为单位.

例:

JF140

LE 命令: 画直线(两直线相交处作”异或”处理)

语法: LEp1, p2, p3, p4<CR>

参数说明:

p1: 设置 X 坐标, 以点(dots)为单位.

p2: 设置 Y 坐标, 以点(dots)为单位.

p3: 设置直线的水平长度, 以点(dots)为单位.

p4: 设置直线的垂直高度, 以点(dots)为单位.

例:

N

LE50, 30, 100, 10

LE100, 20, 5, 110

W1

打印输出:



LO 命令:画直线(两直线相交处作”或”处理)

语法: LOp1, p2, p3, p4<CR>

参数说明:

p1: 设置 X 坐标, 以点(dots)为单位.

p2: 设置 Y 坐标, 以点(dots)为单位.

p3: 设置直线的水平长度, 以点(dots)为单位.

p4: 设置直线的垂直高度, 以点(dots)为单位.

例:


N

LO50, 30, 100, 10

LO100, 20, 5, 110

W1

打印输出:



LS 命令:画斜线

语法: LSp1, p2, p3, p4, p5<CR>

参数说明:

p1: 设置斜线起始 X 坐标, 以点(dots)为单位.

p2: 设置斜线起始 Y 坐标, 以点(dots)为单位.

P3: 设置斜线粗细, 以点(dots)为单位.

P4: 设置斜线终止 X 坐标, 以点(dots)为单位.

P5: 设置斜线终止 Y 坐标, 以点(dots)为单位.

例:

N

LS50, 30, 10, 100, 80

W1

打印输出:



LW 命令: 画白色直线

语法: LWp1, p2, p3, p4<CR>

参数说明:

p1: 设置 X 坐标, 以点 (dots) 为单位.

p2: 设置 Y 坐标, 以点 (dots) 为单位.

p3: 设置直线的水平长度, 以点 (dots) 为单位.

p4: 设置直线的垂直高度, 以点 (dots) 为单位.

例:

N

LE50, 30, 100, 10

LE50, 60, 100, 10

LE50, 90, 100, 10

LE50, 120, 100, 10

LW100, 20, 5, 110

W1

打印输出:



MD 命令: 打印机测纸

语法: MD<CR>

说明: 此命令等同于“校准纸张探测器”功能（长按“进纸”键）

例:

MD

N

T20, 110, 0, 4, 2, 2, N, "LIFE"

W1

N 命令: 清除打印机缓冲内存的内容

语法: N<CR>

参数说明: 当发送新的一张标签内容到打印机前, 建议使用此命令先清空打印机图形缓存里已有的数据内容.

注: 请不要在 *FORM* 里使用 *N* 命令.

O 命令: 设置打印机的工作状态

语法: O[D, P, L, C, N]<CR>

参数说明:

- D: 设置打印机为热感印(热传导)状态.
- P: 设置打印机为连续送纸状态(缺省).
- L: 设置打印机为打印一张标签后, 暂停等待用户确定再打印下一张标签.
(确定方式:1. 按” FEED” 键;2. 在安装剥纸器情况下, 当用户取走标签后自动打印下一张标签).
- C: 设置打印机为安装切纸刀状态.(在安装切刀的情况下)
- N: 设置打印机为安装剥纸器状态.(在安装剥纸器情况下)

例:

O<CR>	;设置打印机为热转印模式, 无切刀, 无剥纸器
OD<CR>	;设置打印机为热传导模式, 无切刀, 无剥纸器
OC<CR>	;设置打印机为热转印模式, 有切刀, 无剥纸器

注意:

1. 切纸刀与剥纸器不能同时安装.
2. 如果打印机状态设置不正确时, 打印机前面板的指示灯将闪烁, 请参考打印机说明书的故障排除章节.

W 命令: 打印标签

语法: Wp1[, p2]<CR>

参数说明:

p1: 打印标签的数量, 取值范围:1—65535.

p2: 每张标签的复制份数, 取值范围: 1—65535.

注: W 命令不能在 FORM 里使用, 而用 WA 命令代替.

例:

```
FK"TEST"  
FS"TEST"  
C0, 6, N, +1, "Enter Start No. :"  
T20, 50, 0, 4, 1, 1, N, "Label: "  
T120, 50, 0, 4, 1, 1, N, C0  
FE  
N  
Q520, 0  
FR"TEST"  
?  
100  
W2, 3
```

打印输出:

Label: 100

Label: 100

Label: 100

Label: 101

Label: 101

Label: 101

WA 命令: 自动打印标签(只能在 FORM 里使用)

语法: WAp1[, p2]<CR>

说明: 在 FORM 里使用此命令时, 当用户输入全部的变量内容, 打印机将立刻开始打印标签.

参数说明:

p1: 打印标签的数量, 取值范围:1—65535.

p2: 每张标签的复制份数, 取值范围: 1—65535.

例:

```
FK"TEST1"  
FS"TEST1"  
C0, 6, N, +1, "Enter Start No. :"  
T20, 50, 0, 4, 1, 1, N, "Label: "  
T120, 50, 0, 4, 1, 1, N, C0  
WA2  
FE  
N  
Q520, 0  
FR"TEST1"  
?  
100
```

打印输出:

Label: 100

Label: 101

Q 命令: 设置标签的高度和定位间隙\黑线\穿孔的高度

语法: Qp1, p2<CR>

参数说明:

p1: 标签的高度, 以点(dots)为单位. 取值范围:0-65535.

缺省为打印机自动检测到的标签高度.

p2: 标签间的定位间隙/黑线/穿孔的高度, 以点(dots)为单位.

取值范围:16-240.

p2 的取值与标签定位方式相关:

间隙模式(GAP MODE):缺省模式, p2 设置为间隙的高度.


穿孔定位属于间隙模式的特例.

黑线模式(BLACK LINE MODE): p2 设置为黑线的高度.

连续纸模式(CONTINUOUS MODE):p2 设置为 0. 这时候, 纸张探测器用来检测纸张是否用尽.

例:

1. 普通间隙定位模式标签:



假设:

$p_1 = 20.0 \text{ mm}$ (160 dots at 200dpi)

$p_2 = 3.0 \text{ mm}$ (24 dots at 200dpi)

Q命令将设置为:

Q160,24

q 命令: 设置标签的宽度

语法: qp1<CR>


参数说明: p1: 标签的宽度, 以点(dots)为单位.

注: 如果紧跟着 q 命令使用 R 命令, 打印机将忽略 q 命令, 而执行 R 命令.

例:

```
N
q250
T20, 30, 0, 2, 1, 1, N, "q command:"
T20, 60, 0, 2, 1, 1, N, "Label width: 250 dots"
W1
```

**Image Buffer
Positioning - Center
Aligned Printers**



R 命令: 设置/改变坐标原点

语法: `Rp1, p2<CR>`

参数说明:

p1: X 坐标移动的距离, 确以点(dots)为单位.

p2: Y 坐标移动的距离, 确以点(dots)为单位.

S 命令: 设置打印机打印速度

语法: Sp1<CR>

参数说明:

p1: 取值范围为 0—6, 10—80

P1 值	速度
0 or 1	1 ips (25 mm/s)
2	2 ips (50 mm/s)
3	3 ips (75 mm/s)
4	4 ips (100 mm/s)
5	5 ips (125 mm/s)
6	6 ips (150 mm/s)

p1值	速度
10	1.0 ips (25 mm/s)
15	1.5 ips (37 mm/s)
20	2.0 ips (50 mm/s)
25	2.5 ips (63 mm/s)
30	3.0 ips (75 mm/s)
35	3.5 ips (83 mm/s)
40	4.0 ips (100 mm/s)
50	5.0 ips (125 mm/s)
60	6.0 ips (150 mm/s)
70	7.0 ips (175 mm/s)
80	8.0 ips (200 mm/s)

U 命令: 打印机器的设置状态

语法: U<CR>

说明: 要求打印出当前打印机设置\工作状态 (自检信息)

例:

U

打印输出:

1、汉字显示:

条码打印机 X1

版本: ZMIN_PLIXX.XX

产品编号: xxxxxxxxxxxx

FLASH.....4M Bytes

RAM.....8M Bytes

已打印标签 (页) ..x

打印总长度 (米) ..x

串口.....9600, N, 8, 1

打印方式.....热转印

纸张探测器.....反射式

碳带探测器.....有效

撕纸模式.....无效

切纸刀.....无效

剥纸器.....无效

当前打印黑度.....10

xxxx, xxxx, xxxx, xxxx

内部字体:

Font 1: ABCabc0123456789

Font 2: AaBbCc0123456789

Font 3: AaBbCc0123456789

Font 4: AaBbCc0123456789

FONT 5: ABC

Font 6 中文宋体: AaBbCc0123456789

2、西文显示

Label Printer X1

Version: ZMIN_PLIXX.XX

PartNumber: xxxxxxxxx

FLASH.....4M Bytes

RAM.....8M Bytes

Labels(Page).....x

Printed Length(M).x

COM.....9600 ,N, 8, 1

Print Method.....THERMAL TRANSFER

Media Sensor.....Reflect

Ribbon Sensor.....Enable

Tear Off.....Disable

Cutter.....Disable

Peeler.....Disable

Darkness.....10

xxxx, xxxx, xxxx, xxxx

Internal Fonts:

Font 1: ABCabc0123456789

Font 2: AaBbCc0123456789

Font 3: AaBbCc0123456789

Font 4: AaBbCc0123456789

FONT 5: ABC

Font 6 中文宋体: AaBbCc0123456789

UN/US 命令: 取消/设置错误反馈

语法: UN<CR>

US<CR>

说明: 打印机的反馈数据从 RS232 串口返回电脑.

如果打印中发生错误, 打印机将先发送一个 NACK (15H) 字符回电脑, 跟着发送出错编号. 如果没有错误发生, 打印机将在接收到 P 命令后发送 ACK (06H) 字符.

错误代码	说明
0x00	No Error
0x01	Object Exceeded Label Border
0x02	Bar Code Data Length Error
0x03	Insufficient Memory to Store Data
0x04	Memory Configuration Error
0x05	RS-232 Interface Error
0x06	Paper or Ribbon Empty
0x07	Duplicate Name: Form, Graphic or Soft Font
0x08	Name Not Found: Form, Graphic or Soft Font
0x09	Not in Data Entry Mode
0x0a	Print Head Up (Open)
0x0b	Pause Mode or Paused in Peel mode
0x0c	Does not fit in area specified
0x0d	Data length to long
0x0c	PDF-417 coded data to large to fit in bar code
0x0d	
0x0e	

V 命令: 定义变量

语法: Vp1, p2, p3, " MSG" <CR>

说明: 在 FORM 里使用此命令来定义一个变量.

参数说明:

p1: 变量 ID 号码, 取值范围: 00—99.

p2: 最大字符个数, 取值范围: 1—99;

如果使用 KDU, 只能在 16 以内.

p3: 对齐方式; L—左对齐, R—右对齐, C—居中, N—不对齐.

“MSG”: 提示内容, 该内容将会在 KDU 或打印机的显示屏上显示.

例:

```
N
FK"TEST2"
FS"TEST2"
V0, 16, L, "Enter Title:"
C0, 6, N, +1, "Enter Code:"
T100, 100, 0, 4, 1, 1, N, V0
T400, 100, 0, 4, 1, 1, N, C0
FE
```

将以上命令存储到打印机里, 再发送以下命令到打印机

```
Q500, 0
FR"TEST2"
?
Part Number:
1234
W1, 2
```

打印输出:

Part Number : 1234

Part Number : 1234

X 命令: 画矩形

语法: Xp1, p2, p3, p4, p5<CR>

参数说明:

- p1: 起始点的 X 坐标, 以点 (dots) 为单位.
- p2: 起始点的 Y 坐标, 以点 (dots) 为单位.
- P3: 边框的粗细, 以点 (dots) 为单位.
- P4: 终止点的 X 坐标, 以点 (dots) 为单位.
- P5: 终止点的 Y 坐标, 以点 (dots) 为单位.

例:

N

T50, 30, 0, 4, 1, 1, R, "BOXES"


X50, 120, 5, 250, 150

X120, 100, 3, 180, 280

W1

打印输出:

BOXES



Z 命令: 设置标签打印方向

语法: Zp1<CR>

说明: 此命令将改变整张标签上所有内容, 如文本, 条码, 直线, 矩形.

参数说明:

p1: 方向;取值为 B 或 T. 缺省值为 T.

B: 将从标签右下角开始打印.

T: 从标签左上角开始正常打印.

例:

N

ZT

T50, 30, 0, 4, 1, 1, R, "ZT"

W1

ZN/ZS 命令: 取消/选择 FLASH 存储器

语法: ZN<CR>

ZS<CR>

例:

ZS

FK"TEST3"

FS"TEST3"

T100, 100, 0, 4, 1, 1, N, " Test Flash"

FE

ZN

以上命令将存储到打印机的 FLASH 存储器里.

以下命令将执行表格 (FORM): "TEST3"

FR" TEST3"

W1

?命令：下载变量或序列号变量

语法：?<CR>

说明：打印机收到这个命令后, 进入接收变量/计数器赋值状态, 这时, 将依次按照 FORM 里声明变量/计数器的顺序先后赋值, 并且紧跟着?命令之后的每一行内容只能赋给一个变量/计数器, 所以用户输入内容行数必须与声明的变量/计数器个数相等.

FK"form1"	; 删除名称为 “form1” 的表格
FS"form1"	; 开始存储名称为 “form1” 的表格
V00,15,N,"Enter Part Name:"	
V01,5,N,"Enter Quantity:"	
T50,10,0,3,1,1,N,V00	
T50,400,0,3,1,1,N,"Quantity: "V01	
FE	; 结束存储名称为 “form1” 的表格
FR"form1"	; 执行 “form1”
?	; 开始下载变量
Screws	; 变量0 (V00) 的内容
235	; 变量1 (V01) 的内容
W1	; 打印标签

^@命令：将打印机复位

语法：^@<CR>

说明：这个命令对打印机复位，会将打印机设置恢复到出厂状态

不能在打印指令序列的开头或者中间使用此指令，否则会将该指令后面的内容清空，导致打印机不执行该指令后面的指令。

例：

N

T20, 110, 0, 4, 2, 2, N, "LIFE"

W1

^@

^ee 命令: 要求打印机立刻反馈错误报告

语法: ^ee<CR>

说明: 用户可以使用此命令立刻确定打印机的当前错误状态, 打印机将传回 4 个字节到主机:

XX<CR><LF> XX为打印机当前状态代码

状态代码	解释
00	无错误
01	语法错误
82	碳带探测出错
83	标签探测出错
86	切刀检测出错
87	打印头未关闭
88	暂停状态
99	其它错误

TD 命令: 设置打印日期格式

语法: TDp1|p2|p3<CR>

参数说明:

p1, p2, p3: 设置年、月、日格式参数, 见下表:

状态代码	解释
y2	显示 2 位年格式 (10)
y4	显示 4 位年格式 (2010)
me	显示 3 位 ASCII 月份格式 (JAN)
mn	显示 2 位数字月份格式 (01)
dd	显示 2 位数字日格式 (15)

|: 设置显示的分隔符, 如 “-”、“,” “/” 等。

默认打印格式为: y4-mn-dd

例: 设现在是 2010 年 3 月 11 日

TD	//设置默认打印格式: 2010-03-11
TDy2/me/dd	//设置打印格式: 10/Mar/11
TDdd-me-y4	//设置打印格式: 11-Mar-2010
TDdd,mn,y4	//设置打印格式: 11,03,2010

TDdd/me/y4	//设置打印格式
A100,200,0,4,1,2,N,"生产日期"TD	: 生产日期 11/Mar/2010
A100,400,0,4,1,2,N,"下周"TD+7	: 下周 18/Mar/2010
TD	//设置默认的打印格式
A100,400,0,4,1,2,N,"有效日期"TD+30	: 有效日期 2010-04-12

注意:

1. 此指令只对具有实时时钟 (RTC) 功能的机型有效。
2. 使用此指令打印日期时, 打印的日期值受其前面最近一条打印时间指令 (TT) 相关联, 具体参考 TT 指令。

TS 命令：设置打印机内部 RTC 日期和时间

语法：TS p1,p2,p3,p4,p5,p6

参数说明：

- p1 : 月份设置，取值范围：01-12；
- p2 : 日设置，取值范围：01-31；
- p3 : 年设置，取值范围：00-99；代表年度 2000-2099
- p4 : 小时设置，取值范围：00-23；
- p5 : 分钟设置，取值范围：00-59；
- p6 : 秒钟设置，取值范围：00-59；

例：

TS03, 11, 10, 13, 20, 00 //RTC 日期被设置为 2010 年 3 月 11 日 13 点 20 分
TS12, 31, 00, 01, 30, 00 //RTC 日期被设置为 2000 年 12 月 31 日 1 点 30 分

注意：

1. 此指令只对具有实时时钟（RTC）功能的机型有效。
2. 打印日期时，打印出的日期与其上面最近一条时间指令相关联。



TT 命令: 设置打印时间格式

语法: TTp1[,p2,p3][+]<CR>

参数说明:

p1, p2, p3=h, m 或 s 设置年、月、日格式参数, 见下表:

状态代码	解释
h	显示 2 位小时格式 (e.g. 11)
m	显示 2 位分钟格式 (e.g. 30)
s	显示 2 位秒钟格式 (e.g. 50)

[+]: 设置 12 小时制模式有效, 使用时, 打印出的时间为 12 小时制, 时间后面带有 “AM” 或 “PM” 标志。默认 (无 “+”) 则打印格式为 24 小时制模式。

默认打印格式为: h:m:s

例: 设现在是 2010 年 3 月 11 日 22 点 50 分 30 秒

TT	//设置默认时间格式: 22: 50: 30
TTh,m,s	//设置时间格式: 22: 50: 30
TTh,m	//设置时间格式: 22: 50
TTh:m:s+	//设置时间格式: 10:50:30PM
TTh:m:s+	//设置时间格式: 10:50:30PM
A100,200,0,4,1,2,N,"现在是"TT	现在是 10: 50: 30PM
A100,400,0,4,1,2,N,"一小时后是"TT+60	一小时后是 11: 50: 30PM

注意:

- 1. 此指令只对具有实时时钟 (RTC) 功能的机型有效。
- 2. TD 指令和 TT 指令同时使用时, 打印的 TD 日期值由上面最近一条的 TT 打印值相关联。

如: 设现在是 2010 年 3 月 11 日 23 点 55 分 30 秒

TTh:m+		
TDdd-mn-y4		
A100,200,0,4,1,2,N,"一小时后"TT+60	一小时后 00: 50AM	//TT+60 (当前时间加 60min) 的时间
A100,300,0,4,1,2,N,"日期"TD	日期 2010-03-12	//TT+60 相应的日期值
A100,400,0,4,1,2,N,"日期"TD+1	日期 2010-03-13	//TT+60 再加 1 天的日期值
A100,400,0,4,1,2,N,"日期"TD+2	日期 2010-03-14	//TT+60 再加 2 天的日期值
A100,500,0,4,1,2,N,"现在时间"TT	现在时间 23: 55PM	//打印 TT(当前时间)日期值
A100,600,0,4,1,2,N,"现在日期"TD	现在日期 2010-03-11	//TT (当前时间) 对应的日期值

